

Quantile regression through linear programming

Anton Antonov

Mathematica for Prediction blog

Mathematica for Prediction project at GitHub

December 2013

Introduction

We can say that least squares linear regression corresponds to finding the mean of a single distribution. Similarly, quantile regression corresponds to finding quantiles of a single distribution. With quantile regression we obtain curves -- called "regression quantiles" -- that together with the least squares regression curve would give a more complete picture of the distributions (the y 's) corresponding to a set of x 's.

For a complete, interesting, and colorful introduction and justification to quantile regression see [2]. An introduction and description of the major properties of quantile regression is given in the Wikipedia entry [3].

In order to have a fast enough for practical purposes implementation of quantile regression we need to re-cast the quantile regression problem as linear programming problem. (Such a formulation is also discussed in [2].)

This document is mostly a guide for usage of the *Mathematica* package for quantile regression that is provided by the MathematicaForPrediction project at GitHub, see [1].

The second section provides theoretical background of the linear programming formulation of the quantile regression problem. The third section shows examples of finding regression quantiles using the function `QuantileRegression` provided by [1]. The last section describes profiling experiments and their results.

The motivational examples in the theoretical section, formulas (1) and (2), can be completed with more expansions and proofs. (Which will be done in the next version of the document.)

Theory

We can formulate the quantile regression problem in way analogous to the formulation of least squares (conditional mean) regression.

Consider a random variable Y having some distribution function F and a sample $\{y_i\}_{i=1}^n$ of Y . The median of the set of samples $\{y_i\}_{i=1}^n$ can be defined as the solution of the minimization problem

$$\min_{\beta} \sum_{i=1}^n |y_i - \beta|, \quad \beta \in \mathbb{R}. \quad (1)$$

To see that the β which minimizes (1) is the median of $\{y_i\}_{i=1}^n$, consider two points $y_1 < y_2$. Then $|y_1 - \mu| + |\mu - y_2| = |y_1 - y_2|$, $\forall \mu \in [y_1, y_2]$, hence any $\mu \in [y_1, y_2]$ minimizes (1). Using the observation for two points we see that for three points $y_1 < y_2 < y_3$, $\mu = y_2$ minimizes (1). For four points $y_1 < y_2 < y_3 < y_4$ any $\mu \in [y_2, y_3]$ minimizes (1). We can generalize these observations and show that (1) gives the median for any set of points.

If we want to find θ -th sample quantile of $\{y_i\}_{i=1}^n$ then we need to change (1) into

$$\min_{\beta} \left(\sum_{y_i \geq \beta} \theta |y_i - \beta| + \sum_{y_i < \beta} (1 - \theta) |y_i - \beta| \right), \quad \beta \in \mathbb{R}. \quad (2)$$

Consider a set of random variables Y_i , $i \in [1, n]$, $n \in \mathbb{N}$ that are paired with a set of x -coordinates $X = \{x_i\}_{i=1}^n$. We have data of pairs $\{x_i, y_i\}_{i=1}^n$, where y_i is a realization of Y_i .

The linear regression problem can be formulated as

$$\min_{\beta_0, \beta_1} \sum_{i=1}^n (y_i - (\beta_0 + \beta_1 x_i))^2. \quad (3)$$

Similarly, the median regression problem can be formulated as

$$\min_{\beta_0, \beta_1} \sum_{i=1}^n |y_i - (\beta_0 + \beta_1 x_i)|. \quad (4)$$

and the θ -th quantile regression problem as

$$\min_{\beta_0, \beta_1} \left(\sum_{i \in \{i: y_i \geq \beta_0 + \beta_1 x_i\}} \theta |y_i - (\beta_0 + \beta_1 x_i)| + \sum_{i \in \{i: y_i < \beta_0 + \beta_1 x_i\}} (1 - \theta) |y_i - (\beta_0 + \beta_1 x_i)| \right), \quad \beta_0, \beta_1 \in \mathbb{R}. \quad (5)$$

In order to convert (5) into a linear programming problem, let us introduce the non-negative variables u_i and v_i for which the following equations are true:

$$\begin{aligned} y_i - (\beta_0 + \beta_1 x_i) + u_i &= 0, \quad i \in \{i: y_i \geq \beta_0 + \beta_1 x_i\}, \\ u_i &= 0, \quad i \notin \{i: y_i \geq \beta_0 + \beta_1 x_i\}, \end{aligned} \quad (6)$$

$$\begin{aligned} (\beta_0 + \beta_1 x_i) - y_i + v_i &= 0, \quad i \in \{i: y_i < \beta_0 + \beta_1 x_i\}, \\ v_i &= 0, \quad i \notin \{i: y_i < \beta_0 + \beta_1 x_i\}. \end{aligned} \quad (7)$$

Since we have $u_i \geq 0$, $v_i \geq 0$, $\forall i \in [1, n]$ we can re-write (6) and (7) simply as

$$y_i - (\beta_0 + \beta_1 x_i) + u_i = 0, \quad u_i \geq 0, \quad i \in [1, n], \quad (8)$$

$$y_i - (\beta_0 + \beta_1 x_i) - v_i = 0, \quad v_i \geq 0, \quad i \in [1, n]. \quad (9)$$

Then (5) expressed with u_i and v_i becomes

$$\min_{u_i, v_i, \beta_0, \beta_1} \left(\sum_{i \in \{i: y_i \geq \beta_0 + \beta_1 x_i\}} \theta u_i + \sum_{i \in \{i: y_i < \beta_0 + \beta_1 x_i\}} (1 - \theta) v_i \right). \quad (10)$$

Since $u_i \geq 0$ and $v_i \geq 0$ the minimization function (10) can simply be written as

$$\min_{u_i, v_i, \beta_0, \beta_1} \left(\sum_{i=1}^n \theta u_i + \sum_{i=1}^n (1 - \theta) v_i \right). \quad (11)$$

The equations (8), (9) and formula (11) are the linear programming formulation of the quantile regression problem (5).

Note that $u_i v_i = 0, \forall i \in [1, n]$.

The quantile regression formulations (5), and (8), (9) and (11) can be done for any model of Y_i that is a linear combination of functions over X not just for the linear model $\beta_0 + \beta_1 X$.

Examples of usage

Package load


Load the package [1]:

```
In[94]:= Get["~/MathFiles/MathematicaForPrediction/QuantileRegression.m"]
```

Logarithmic curve with noise

Let us generate some data.

```
In[239]:= Block[{n = 1200, start = 10, end = 200},
  data = Table[{t, 5 + Log[t] +
    RandomReal[SkewNormalDistribution[0, Log[t] / 5, 12]]},
    {t, Rescale[Range[1, n], {1, n}, {start, end}]}];
ListPlot[data, AxesLabel → {"x", "y"},
  PlotRange → All, ImageSize → 400]
```



Consider the following quantiles:

```
In[241]:= qs = {0.05, 0.25, 0.5, 0.75, 0.95};
```

We want to find curves that separate the data according the quantiles. Those curves are called “regression quantiles”.

Pretending that we do not know how the data is generated, just by looking at the plot we assume that the model for the data is

$$y = \beta_0 + \beta_1 x + \beta_2 \sqrt{x} + \beta_3 \log(x). \quad (12)$$

Let us put the model functions for the regression fit in the variable `funcs`:

```
In[242]:= funcs = {1, x,  $\sqrt{x}$ , Log[x]};
```

Here we find the regression quantiles:

```
In[243]:= qrFuncs = QuantileRegression[data, funcs, x, qs];  
TableForm[List /@ qrFuncs]
```

Out[244]//TableForm=


```
4.98984 + 0.00360806  $\sqrt{x}$  +  $4.41887 \times 10^{-12} x$  + 1.00758 Log[x]  
5.05758 + 0.0127716  $\sqrt{x}$  +  $8.03872 \times 10^{-14} x$  + 1.01985 Log[x]  
5.3324 +  $8.48918 \times 10^{-6} \sqrt{x}$  + 0.00143325 x + 1.02662 Log[x]  
5.21087 +  $4.70418 \times 10^{-13} \sqrt{x}$  +  $2.40149 \times 10^{-12} x$  + 1.17793 Log[x]  
5.31292 +  $5.34391 \times 10^{-10} \sqrt{x}$  +  $3.2465 \times 10^{-10} x$  + 1.3126 Log[x]
```

We also apply `Fit` to the data and the model functions in order to compare the regression quantiles with the least-squares regression fit:

```
In[245]:= fFunc = Fit[data, funcs, x]
```

Out[245]= $4.96969 - 0.0956781 \sqrt{x} + 0.003187 x + 1.2993 \text{Log}[x]$

Here is a plot that combines the found regression quantiles and least squares fit:



Let us check how good the regression quantiles are for separating the data according to the quantiles they were computed for:

```
In[249]:= tbl = Table[
  {qs[[i]], Length[Select[data, #[[2]] ≥ (qrFuncs[[i]] /. x → #[[1]]) &]] /
    Length[data] // N}, {i, Length[qs]}};
TableForm[tbl, TableHeadings → {None,
  {"quantile", "fraction\nabove"}}]
```


Out[250]//TableForm=

quantile	fraction above
0.05	0.951667
0.25	0.749167
0.5	0.499167
0.75	0.25
0.95	0.05

Robustness

Let us demonstrate the robustness of the regression quantiles with the data of the previous example. Suppose that for some reason all the data y -values greater than 11.25 are altered by multiplying them with a some greater than 1 factor, say, $\alpha = 1.2$. Then the altered data looks like this:

```
In[251]:= α = 1.2;
dataAlt = Map[If[#[[2]] > 11.25, {#[[1]], α#[[2]]}, #] &, data];
ListPlot[dataAlt, AxesLabel → {"x", "y"},
  PlotRange → All, ImageSize → 400]
```



Let us compute the regression quantiles for the altered data:

```
In[254]:= qrFuncsAlt = QuantileRegression[dataAlt, funcs, x, qs];  
TableForm[List /@ qrFuncs]
```

Out[255]//TableForm=

$$4.98984 + 0.00360806 \sqrt{x} + 4.41887 \times 10^{-12} x + 1.00758 \log[x]$$

$$5.05758 + 0.0127716 \sqrt{x} + 8.03872 \times 10^{-14} x + 1.01985 \log[x]$$

$$5.3324 + 8.48918 \times 10^{-6} \sqrt{x} + 0.00143325 x + 1.02662 \log[x]$$

$$5.21087 + 4.70418 \times 10^{-13} \sqrt{x} + 2.40149 \times 10^{-12} x + 1.17793 \log[x]$$


$$5.31292 + 5.34391 \times 10^{-10} \sqrt{x} + 3.2465 \times 10^{-10} x + 1.3126 \log[x]$$

and let us also compute the least squares fit of the model (1):

```
In[256]:= fFuncAlt = Fit[dataAlt, funcs, x]
```

Out[256]= $4.31095 - 0.949037 \sqrt{x} + 0.0378394 x + 2.569 \log[x]$

Here is a plot that combines the functions found over the altered data:



We can see that the new regression quantiles computed for 0.05, 0.25, and 0.5 have not changed significantly:

```
In[260]:= qrFuncs[[1 ;; 3]]
```

Out[260]= $\left\{ 4.98984 + 0.00360806 \sqrt{x} + 4.41887 \times 10^{-12} x + 1.00758 \log[x], \right.$
 $5.05758 + 0.0127716 \sqrt{x} + 8.03872 \times 10^{-14} x + 1.01985 \log[x],$
 $5.3324 + 8.48918 \times 10^{-6} \sqrt{x} + 0.00143325 x + 1.02662 \log[x] \left. \right\}$

```
In[261]:= qrFuncsAlt[[1 ;; 3]]
```

Out[261]= $\left\{ 4.98984 + 0.00360806 \sqrt{x} + 2.15719 \times 10^{-13} x + 1.00758 \log[x], \right.$
 $5.05758 + 0.0127716 \sqrt{x} + 4.12893 \times 10^{-11} x + 1.01985 \log[x],$
 $5.33238 + 2.50975 \times 10^{-10} \sqrt{x} + 0.0014335 x + 1.02664 \log[x] \left. \right\}$

ant that they are still good for separating the un-altered data:

```
In[262]:= tbl = Table[
  {qs[[i]], Length[Select[data, #[[2]] ≥ (qrFuncsAlt[[i]] /. x → #[[1]]) &]] /
    Length[data] // N}, {i, Length[qs]}; TableForm[tbl,
  TableHeadings → {None, {"quantile", "fraction\nabove"}}]
```

Out[262]//TableForm=

quantile	fraction above
0.05	0.949167
0.25	0.75
0.5	0.500833
0.75	0.186667
0.95	0.0108333

Also we can see that the least squares fit of (1) has significantly changed:

```
In[263]:= fFunc
```


Out[263]= $4.96969 - 0.0956781 \sqrt{x} + 0.003187 x + 1.2993 \log[x]$

```
In[264]:= fFuncAlt
```

Out[264]= $4.31095 - 0.949037 \sqrt{x} + 0.0378394 x + 2.569 \log[x]$

Data generated with Sin and noise

```
In[265]:= With[{n = 2000, start = 10, end = 140},
  data = Table[{t,
    Sin[10 + t π / 15] + RandomReal[NormalDistribution[0, Log[t] / 3]]},
    {t, Rescale[Range[1, n], {1, n}, {start, end}]}];
  data[[All, 2]] = data[[All, 2]] + Log[data[[All, 1]]];
]
ListPlot[data, PlotRange → All, ImageSize → 400]
```



We are going to use again the quantiles:

```
In[221]:= qs = {0.05, 0.25, 0.5, 0.75, 0.95};
```

We can derive a guess about the data model by observing that we have peaks at $x = 50$, $x = 80$, $x = 110$. With these observations we make equations for Solve and take the second solution:

```
In[222]:= Solve[{a > 0, Sin[x α] == Sin[α (x + 60)], Sin[x α] == Sin[α (x + 30)]}, α]
```

```
Out[222]:= Solve[{a > 0, Sin[1.2 x] == Sin[1.2 (60 + x)],  
Sin[1.2 x] == Sin[1.2 (30 + x)]}, 1.2]
```

```
In[223]:= π / 15 // N
```

```
Out[223]:= 0.20944
```

Next we need to find a guess for the phase. Again we use the second solution provided by Solve:

```
In[224]:= Solve[{φ > 0, Sin[φ + 50 π / 15] == 1}, φ]
```

```
Out[224]:= { {φ → ConditionalExpression[  
1/6 (-5 π - 12 π C[1]), C[1] ∈ Integers && C[1] ≤ -1] },  
{φ → ConditionalExpression[1/6 (7 π - 12 π C[1]),  
C[1] ∈ Integers && C[1] ≤ 0] } }
```

```
1/6 7 π // N
```

```
3.66519
```


Alternatively, we can simply use Manipulate and plot the data together with a model function subject to different parameters change.

```

In[167]:= Manipulate[
  DynamicModule[{gr1, gr2},
    gr1 = ListPlot[data, PlotRange -> All];
    gr2 = Plot[α Sin[φ + β x], {x, 0, 140}, PlotStyle -> Darker[Red]];
    Show[{gr1, gr2}]
  ], {{α, 1}, 0.5, 10, 1}, {β, 0, 2, 0.01}, {φ, 0, 30, 0.25}]

```

Out[167]=



From the calculations we did so far we assume that the model for the data is

$$y = \beta_0 + \beta_1 x + \beta_2 \sin[3.7 + x \pi / 15]$$

Let us put the model functions for the regression fit in the variable `funcs`:

```

In[267]:= funcs = {1, x, Sin[3.7 + x π / 15]};

```

We find the regression quantiles:

```
In[268]:= qrFuncs = QuantileRegression[data, funcs, x, qs];
Grid[List /@ qrFuncs]
```


```
1.2696 + 0.00761991 x + 0.700591 Sin[3.7 +  $\frac{\pi x}{15}$ ]
2.15376 + 0.013776 x + 0.997599 Sin[3.7 +  $\frac{\pi x}{15}$ ]
Out[269]= 2.89702 + 0.0168659 x + 1.02199 Sin[3.7 +  $\frac{\pi x}{15}$ ]
3.54742 + 0.0207752 x + 1.01092 Sin[3.7 +  $\frac{\pi x}{15}$ ]
4.42559 + 0.0263984 x + 0.79197 Sin[3.7 +  $\frac{\pi x}{15}$ ]
```

As in the previous example we also apply `Fit` to the data and the model functions in order to compare the regression quantiles with the least-squares regression fit:

```
In[270]:= fFunc = Fit[data, funcs, x]
```

```
Out[270]= 2.91364 + 0.0163253 x + 0.962757 Sin[3.7 +  $\frac{\pi x}{15}$ ]
```

Here is a plot that combines the functions found:



Let us check how good the regression quantiles are:

```
In[232]:= tbl = Table[
  {qs[[i]], Length[Select[data, #[[2]] ≥ (qrFuncs[[i]] /. x → #[[1]]) &]] /
    Length[data] // N}, {i, Length[qs]}};
TableForm[tbl, TableHeadings → {None,
  {"quantile", "fraction\nabove"}}]
```

Out[233]//TableForm=

quantile	fraction above
0.05	0.9495
0.25	0.75
0.5	0.5005
0.75	0.25
0.95	0.0505

Profiling

It is interesting to see the timing profile of the computations with `QuantileRegression` across two axes: (i) data size and (ii) number of functions to be fit.

First we need to choose a family or several families of test data. Also, since *Mathematica*'s function `LinearProgramming` has several methods it is a good idea to test with all of them. Here I am going to show results only with one family of data and two `LinearProgramming` methods. The data family is the skewed noise over a logarithmic curve used as an example above. The first `LinearProgramming` method is *Mathematica*'s (default) "InteriorPoint", the second method is "CLP" that uses the built-in COIN-OR CLP optimizer. I run the profiling tests using one quantile {0.5} and five quantiles {0.05, 0.25, 0.5, 0.75, 0.95}, which are shown in blue and red respectively. I also run tests with different number of model functions $\{1, x, \sqrt{x}, \text{Log}[x]\}$ and $\{1, x, \text{Log}[x]\}$ but there was no significant difference in the timings (less than 2%).

Test family functions definitions

In this sub-section are shown definitions of functions generating families of data sets.

```
Clear[LogarithmicCurveWithNoise]
LogarithmicCurveWithNoise[
  nPoints_Integer, start_?NumberQ, end_?NumberQ] :=
Block[{data},
  data = Table[{t, 5 + Log[t] +
    RandomReal[SkewNormalDistribution[0, Log[t] / 5, 12]]},
    {t, Rescale[Range[1, nPoints], {1, nPoints}, {start, end}]}];
  data
];

Clear[SinWithUpwardTrend]
SinWithUpwardTrend[
  nPoints_Integer, start_?NumberQ, end_?NumberQ] :=
Block[{data},
  data = Table[{t,
    Sin[t  $\pi$  / 15] + RandomReal[NormalDistribution[0, Log[t] / 3]]},
    {t, Rescale[Range[1, nPoints], {1, nPoints}, {start, end}]}];
  data[[All, 2]] = data[[All, 2]] + data[[All, 1]]^1 / 6;
  data
];
```

```
In[171]:= Clear[SinWithParabolaTrend]
SinWithParabolaTrend[
  nPoints_Integer, start_?NumberQ, end_?NumberQ] :=
Block[{data},
  data = Table[
    {t, Sin[t  $\pi$  / 10] + RandomReal[NormalDistribution[0,  $\sqrt{t}$  / 10]]},
    {t, Rescale[Range[1, nPoints], {1, nPoints}, {start, end}]}];
  data[[All, 2]] = data[[All, 2]] +
    (Mean[{start, end}] - data[[All, 1]])^2 / 300;
  data
];
```

LogarithmicCurveWithNoise 4 model functions with Method→ LinearProgramming

```
modelFuncs = {1, x,  $\sqrt{x}$ , Log[x]};

qs = {0.05`, 0.25`, 0.5`, 0.75`, 0.95`}
{0.05, 0.25, 0.5, 0.75, 0.95}

dataSets = LogarithmicCurveWithNoise[#, 10., 100.] & /@
  Range[500, 10 000, 500];
dataSets // Length
20


timingsLogarithmicCurveWithNoiseF4Q1 =
  Map[{Length[#], AbsoluteTiming[QuantileRegression[#, modelFuncs, x,
    {0.5}, Method → LinearProgramming];][[1]]} &, dataSets]
{{500, 0.098310}, {1000, 0.298962},
 {1500, 0.588580}, {2000, 0.987977}, {2500, 1.412103},
 {3000, 1.961045}, {3500, 2.658844}, {4000, 3.302903},
 {4500, 3.824936}, {5000, 4.812682}, {5500, 5.670178},
 {6000, 7.059653}, {6500, 8.519677}, {7000, 9.950919},
 {7500, 10.886583}, {8000, 12.807758}, {8500, 14.694095},
 {9000, 16.205932}, {9500, 17.815996}, {10 000, 21.027593}}
```

```

timingsLogarithmicCurveWithNoiseF4Q5 =
  Map[{Length[#], AbsoluteTiming[QuantileRegression[#, modelFuncs,
    x, qs, Method → LinearProgramming];][[1]]} &, dataSets]
{{500, 0.549529}, {1000, 1.494088},
 {1500, 2.591068}, {2000, 3.493207}, {2500, 5.204162},
 {3000, 6.551883}, {3500, 7.290538}, {4000, 7.828214},
 {4500, 11.299679}, {5000, 12.629508}, {5500, 13.832353},
 {6000, 18.640458}, {6500, 19.818611}, {7000, 21.795818},
 {7500, 23.999628}, {8000, 27.048659}, {8500, 30.410464},
 {9000, 33.484113}, {9500, 38.420541}, {10000, 38.569162}}

```

QuantileRegression[___, Method → LinearProgramming]
 timings per data size with four model functions
 for skewed noise over a logarithmic curve




```

Mean[timingsLogarithmicCurveWithNoiseF4Q1[[All, 2]] /
timingsLogarithmicCurveWithNoiseF4Q5[[All, 2]]]
0.380121

```

```
ListPlot[Transpose[{timingsLogarithmicCurveWithNoiseF4Q1[All, 1],
  timingsLogarithmicCurveWithNoiseF4Q1[All, 2] /
  timingsLogarithmicCurveWithNoiseF4Q5[All, 2]}],
  PlotStyle -> {PointSize[0.02]}, PlotRange -> All]
```



LogarithmicCurveWithNoise 4 model functions with Method -> {LinearProgramming, Method -> "CLP"}

The same model functions and quantiles were used as in the previous sub-section.


```
timingsLogarithmicCurveWithNoiseCLPF4Q1 =
  Map[{Length[#], AbsoluteTiming[
    QuantileRegression[#, modelFuncs, x, {0.5}, Method ->
      {LinearProgramming, Method -> "CLP"}];][[1]] &, dataSets]

{{500, 0.057988}, {1000, 0.196983},
 {1500, 0.419909}, {2000, 0.795889}, {2500, 1.187998},
 {3000, 1.706534}, {3500, 2.287483}, {4000, 2.910575},
 {4500, 3.583797}, {5000, 4.425360}, {5500, 5.352640},
 {6000, 6.446963}, {6500, 8.402463}, {7000, 9.487401},
 {7500, 10.838176}, {8000, 12.300428}, {8500, 14.839014},
 {9000, 15.998385}, {9500, 17.698214}, {10000, 20.404420}}

timingsLogarithmicCurveWithNoiseCLPF4Q5 = Map[{Length[#],
  AbsoluteTiming[QuantileRegression[#, modelFuncs, x, qs, Method ->
    {LinearProgramming, Method -> "CLP"}];][[1]] &, dataSets]


{{500, 0.120797}, {1000, 0.411864},
 {1500, 0.913235}, {2000, 1.595715}, {2500, 2.369918},
 {3000, 3.433828}, {3500, 4.666453}, {4000, 6.038684},
 {4500, 7.456060}, {5000, 9.125450}, {5500, 10.848903},
 {6000, 13.103446}, {6500, 15.463412}, {7000, 17.807215},
 {7500, 20.780654}, {8000, 23.596392}, {8500, 26.951688},
 {9000, 31.034202}, {9500, 34.052749}, {10000, 38.549676}}
```


QuantileRegression[___,Method->{LinearProgramming,Method->"CLP"}]
 timings per data size with four model functions
 for skewed noise over a logarithmic curve



```
Mean[timingsLogarithmicCurveWithNoiseCLPF4Q1[All, 2] /
      timingsLogarithmicCurveWithNoiseCLPF4Q5[All, 2]]
0.50362
```

```
ListPlot[
  Transpose[{timingsLogarithmicCurveWithNoiseCLPF4Q1[All, 1],
             timingsLogarithmicCurveWithNoiseCLPF4Q1[All, 2] /
             timingsLogarithmicCurveWithNoiseCLPF4Q5[All, 2]}],
  PlotStyle -> {PointSize[0.02]}, PlotRange -> All]
```



Notes

It is interesting to note that the average ratio of the timings with 1 vs. 5 quantiles is 0.38 for "InteriorPoint" and 0.5 for "CLP".

Careful with “CLP”

During the profiling experiments with some of data families was observed that “CLP” gives curves that do not follow the data closely. Below is shown such a computation with both “InteriorPoint” and “CLP”, the former looks good, the latter does not.

```
In[168]:= modelFuncs =
           {1, x, Sin[x π / 10], -Sin[x π / 10], -Sin[x π / 200], Sin[x π / 200]};

In[169]:= qs = {0.05`, 0.25`, 0.5`, 0.75`, 0.95`}
Out[169]= {0.05, 0.25, 0.5, 0.75, 0.95}


In[181]:= dataSet = SinWithParabolaTrend[1500, 10., 100.];

In[182]:= AbsoluteTiming[
           qrFuncs = QuantileRegression[dataSet,
           modelFuncs, x, qs, Method → {LinearProgramming}]
           ]
Out[182]= {4.363524, {7.74411 + 0.3587 x - 37.887 Sin[ $\frac{\pi x}{200}$ ] + 0.939716 Sin[ $\frac{\pi x}{10}$ ],
           8.00623 + 0.351127 x - 36.6666 Sin[ $\frac{\pi x}{200}$ ] + 0.908217 Sin[ $\frac{\pi x}{10}$ ],
           8.37742 + 0.359772 x - 37.1517 Sin[ $\frac{\pi x}{200}$ ] + 0.966431 Sin[ $\frac{\pi x}{10}$ ],
           8.67913 + 0.365803 x - 37.2965 Sin[ $\frac{\pi x}{200}$ ] + 1.00878 Sin[ $\frac{\pi x}{10}$ ],
           8.95491 + 0.362633 x - 36.4475 Sin[ $\frac{\pi x}{200}$ ] + 0.94464 Sin[ $\frac{\pi x}{10}$ ]}}}
```

```

In[183]:= grData = ListPlot[dataSet, PlotRange → All];
grFit = Plot[Evaluate[MapThread[Tooltip[#1, #2] &, {qrFuncs, qs}]],
  {x, Min[dataSet[[All, 1]], Max[dataSet[[All, 1]]]},
  PlotStyle → AbsoluteThickness[1.4]];
Show[{grData, grFit}, ImageSize → 500]

```



```

In[185]:= AbsoluteTiming[
  qrFuncs = QuantileRegression[dataSet, modelFuncs, x, qs, Method →
    {LinearProgramming, Method → "CLP", Tolerance → 10^-14.0}]
]

```


Out[185]=

$$\left\{ 1.158118, \left\{ -3.09553 + 3.15342 \sin\left[\frac{\pi x}{200}\right] + 0.968002 \sin\left[\frac{\pi x}{10}\right], \right. \right. \\
 -3.09553 + 4.4509 \sin\left[\frac{\pi x}{200}\right] + 0.835651 \sin\left[\frac{\pi x}{10}\right], \\
 3.97571 - 3.13847 \sin\left[\frac{\pi x}{200}\right] + 0.608836 \sin\left[\frac{\pi x}{10}\right], \\
 8.67913 + 0.365803 x - 37.2965 \sin\left[\frac{\pi x}{200}\right] + 1.00878 \sin\left[\frac{\pi x}{10}\right], \\
 \left. \left. -3.09553 + 22.5929 \sin\left[\frac{\pi x}{200}\right] - 1.91918 \sin\left[\frac{\pi x}{10}\right] \right\} \right\}$$

```

In[186]:= grData = ListPlot[dataSet, PlotRange → All];
grFit = Plot[Evaluate[MapThread[Tooltip[#1, #2] &, {qrFuncs, qs}]],
  {x, Min[dataSet[[All, 1]], Max[dataSet[[All, 1]]]},
  PlotStyle → AbsoluteThickness[1.4]];
Show[{grData, grFit}, ImageSize → 500]

```



References

- [1] Anton Antonov, Quantile regression *Mathematica* package, source code at GitHub, <https://github.com/antononcube/MathematicaForPrediction>, package QuantileRegression.m, (2013).
- [2] Roger Koenker, Gilbert Bassett Jr., “Regression Quantiles”, *Econometrica*, 46(1), 1978, pp. 33-50.
JSTOR URL: <http://links.jstor.org/sici?sici=0012-9682%28197801%2946%3A1%3C33%3ARQ%3E2.0.CO%3B2-J>.
- [3] Wikipedia, Quantile regression, http://en.wikipedia.org/wiki/Quantile_regression.
- [4] Brian Cade, Barry Noon, “A gentle introduction to quantile regression for ecologists”, *Front. Ecol. Environ.* 1(8), 2003, pp. 412–420.